
ΠΑΡΑΡΤΗΜΑ Α (1ου ΜΑΘΗΜΑΤΟΣ): ΠΑΡΑ∆ΕΙΓΜΑΤΑ BUGS 
 
1 EXAMPLE 1: BIRTHWEIGHT & ESTRIOL LEVEL 

1.1 Model (Bug file) 
model example1; 
const n=31; # n=sample size 
var   estriol[n],  # estriol level of pregant woman 
      birth[n],    # birthweight  
      mu[n],       # regression expected value  
      a.star,a,b,tau,s2;  # model parameters, 
                   # tau = precision, s2=1/tau error variance 
data estriol,birth in 'estriol.dat'; 
inits in 'estriol2.ini'; 
{ 
#       definition of likelihood function 
# 
        for (i in 1:n) { 
                birth[i]~dnorm( mu[i], tau ); # random component 
#       systematic component & link function 
                mu[i]<-a.star+b*(estriol[i]-mean(estriol[]));         
        } 
#       prior distributions 
#        
        a.star~dnorm( 0, 1.0E-04 ); # normal prior for a 
        b~dnorm( 0, 1.0E-04 ); # normal prior for b 
        tau~dgamma( 1.0E-04 , 1.0E-04 ); # gamma prior for precision 
        s2<-1/tau; 
        a<-a.star-b*mean(estriol[]);  
} 

1.2 Initial Values (estriol2.ini) 
list(a.star=0.0, b=0.0, tau=1.0)  

1.3 Data (estriol.dat) 
7 25 
9 25 
9 25 
12 27 
14 27 
16 27 
16 24 
14 30 
16 30 
16 31 
17 30 
19 31 
21 30 
24 28 
15 32 
16 32 
17 32 
25 32 
27 34 
15 34 
15 34 
15 35 
16 35 
19 34 
18 35 
17 36 
18 37 
20 38 
22 40 
25 39 

24 43 
 
 

2 EXAMPLE 2: BEETLES DATASET 

2.1 Model (Bug file) 
model beetles; 
const 
 N = 8;    # number of doses 
var  
 r[N],p[N],x[N],n[N],alpha,alpha.star,beta,r.hat[N], odds.ratio; 
data x, n, r in "beetles.dat"; 
inits in "beetles.in"; 
{ 
 for (i in 1:N) { 
   r[i] ~ dbin(p[i], n[i]); 
   logit(p[i])   <- alpha.star + beta*(x[i]-mean(x[]));  
   r.hat[i] <- p[i]*n[i];  # fitted values 
 } 
 alpha.star ~ dnorm(0.0, 1.0E-3); 
 beta      ~ dnorm(0.0, 1.0E-3); 
 alpha     <- alpha.star - beta*mean(x[]); 
 odds.ratio <- exp( beta ) 
} 

2.2 Initial Values (beetles.ini) 
list(alpha.star=0, beta=0) 

2.3 Data (beetles.dat) 
1.6907  59   6 
1.7242  60  13 
1.7552  62  18 
1.7842  56  28 
1.8113  63  52 
1.8369  59  53 
1.8610  62  61 
1.8839  60  60 

3 EXAMPLE 3: WAIS & SENILITY SYMPTOMS 

3.1 Model (Bug file) 
model example3; 
const n=54;       # number of observations 
var   wais[n],    # wais measurement 
      symptom[n], # symptom binary indicator 
      p[n],       # probability of symptom appearance  
      alpha,      # alpha parameter  
      beta,       # beta parameter  
      odds.ratio, # odds ratio 
      x.fifty,    # value of x for p=1/2 
      p.meanx;    # fitted probability for x=mean(x) 
data wais, symptom in 'wais.dat'; 
inits in 'wais.ini'; 
{ 
#       Likelihood Definition 
        for (i in 1:n) { 
#               random component 
                symptom[i]~dbern( p[i] ); 
#               systemantic component & link function 
                logit( p[i] ) <- alpha + beta * wais[i]; 
        } 
#       calculation of odds ratio 
        odds.ratio<-exp(beta); 
#       priors for alpha and beta 
        alpha~dnorm(0.0, 0.00001); 
        beta~dnorm(0.0, 0.00001); 
#       x.fifty 
        x.fifty<- -alpha/beta 
#       symptom probability for mean of observed wais 
        p.meanx<-exp( alpha+beta*mean(wais[]) )/ 
(1+exp(alpha+beta*mean(wais[]) ) ) 
} 

3.2 Initial Values (wais.ini) 
list( alpha=0.0, beta=0.0 ) 

3.3 Data (wais.dat) 
9 1 
13 1 
6 1 
8 1 
10 1 
4 1 
14 1 
8 1 
11 1 
7 1 
9 1 
7 1 
5 1 
14 1 
13 0 
16 0 
10 0 
12 0 
11 0 
14 0 
15 0 
18 0 
7 0 
16 0 

9 0 
9 0 
11 0 
13 0 
15 0 
13 0 
10 0 
11 0 
6 0 
17 0 
14 0 
19 0 
9 0 
11 0 
14 0 
10 0 
16 0 
10 0 
16 0 
14 0 
13 0 
13 0 
9 0 
15 0 
10 0 
11 0 
12 0 
4 0 
14 0 
20 0 

4 EXAMPLE 4: BREAST CANCER & AGE AT 1st BIRTH 
 

4.1 Model (Bug file) 
model example4; 
const n=4;        # number of cells 
var   status[n],  # study group (1=case, 0=control) 
      age[n],     # age at 1st birth (0=age<30, 1=age>29) 
      counts[n],  # cell counts 
      lambda[n],  # expected number of cells 
      mu,         # constant parameter 
      a,          # status effect 
      b,          # age effect 
      ab,         # interactio between status and age 
      odds.ratio; # odds ratio 
data status, age, counts in 'breast.dat'; 
inits in 'breast.ini'; 
{ 
#       Likelihood Definition 
        for (i in 1:n) { 
#               random component 
                counts[i]~dpois( lambda[i] ); 
#               systemantic component & link function 
                log( lambda[i] ) <- mu + a*status[i] +b*age[i]+ 
ab*status[i]*age[i]; 
        } 
#       calculation of odds ratio 
        odds.ratio<-exp(ab); 
#       priors for model parameters 
        mu~dnorm(0.0, 0.00001); 
        a~dnorm(0.0, 0.00001); 
        b~dnorm(0.0, 0.00001); 
        ab~dnorm(0.0, 0.00001); 
} 

4.2 Initial Values (breast.ini) 
list(mu=0,a=0,b=0,ab=0) 

4.3 Data (breast.dat) 
1 1 683 
1 0 2537 
0 1 1498 
0 0 8747 

5 EXAMPLE 5: 2x2x2 COMMON ODDS RATIO ESTIMATION: PASSIVE 
SMOKING & CANCER ADJUSTING FOR SMOKING STATUS 

5.1 ANALYSIS 1: DIFFERENT RISK PER SMOKING STATUS 

5.1.1 Model (Bug File) 
model ex5notcommonOR; 
const n=8, 
      p=4; 
var 
      smoking[n], 
      status[n], 
      passive[n], 
      counts[n], 
      b[2,p], 
      lambda[n],  
      or[2]; 
data smoking, status, passive, counts in 'smoke.dat'; 
inits in 'smoke.ini'; 
{ 
# 
#       model for 1st table (nonsmokers) 
        for (i in 1:4) { 
                counts[i]~dpois( lambda[i] ); 
                log(lambda[i])<- b[1,1] + b[1,2]*status[i] + 
b[1,3]*passive[i] + b[1,4]*status[i]*passive[i]; 
        } 
# 
#       model for 2nd table (smokers) 
        for (i in 5:8) { 
                counts[i]~dpois( lambda[i] ); 
                log(lambda[i])<- b[2,1] + b[2,2]*status[i] + 
b[2,3]*passive[i] + b[2,4]*status[i]*passive[i]; 
        } 
#       priors 
        for (i in 1:2){ 
                for (j in 1:p){ 
                        b[i,j]~dnorm(0.0, 1.0E-04) 
                } 
        } 
#       odds ratios 
        or[1]<-exp(b[1,4]) 
        or[2]<-exp(b[2,4]) 
} 

5.1.2 Initial Values (smoke.ini) 
list(b=c(0,0,0,0,0,0,0,0)) 

5.1.3 Data (smoke.dat) 
0 1 1 120 
0 1 0 111 
0 0 1 80 
0 0 0 155 
1 1 1 161 
1 1 0 117 
1 0 1 130 
1 0 0 124 

5.2 ANALYSIS 2: COMMON RISK PER SMOKING STATUS 

5.2.1 Model (Bug File) 
model ex5commonOR; 
const n=8, 
      p=4; 
var 
      smoking[n], 
      status[n], 
      passive[n], 
      counts[n], 
      b[2,p-1], 
      ab,  
      lambda[n],  
      or; 
data smoking, status, passive, counts in 'smoke.dat'; 
inits in 'smoke2.ini'; 
{ 
# 
#       model for 1st table (nonsmokers) 
        for (i in 1:4) { 
                counts[i]~dpois( lambda[i] ); 
                log(lambda[i])<- b[1,1] + b[1,2]*status[i] + 
b[1,3]*passive[i] + ab*status[i]*passive[i]; 
        } 
# 
#       model for 2nd table (smokers) 
        for (i in 5:8) { 
                counts[i]~dpois( lambda[i] ); 
                log(lambda[i])<- b[2,1] + b[2,2]*status[i] + 
b[2,3]*passive[i] + ab*status[i]*passive[i]; 
        } 
#       priors 
        for (i in 1:2){ 
                for (j in 1:p-1){ 
                        b[i,j]~dnorm(0.0, 1.0E-04); 
                } 
        } 
        ab~dnorm(0.0, 1.0E-04); 
#       odds ratios 
        or<-exp(ab) 
} 

5.2.2 Initial Values (smoke2.ini) 
list(b=c(0,0,0,0,0,0),ab=0) 

 
 
 


